# Maintainer: Jan Alexander Steffens (heftig) <heftig@archlinux.org>

pkgname=js102
pkgver=102.15.0
pkgrel=1
pkgdesc="JavaScript interpreter and libraries - Version 102"
url="https://spidermonkey.dev/"
arch=(loong64 x86_64)
license=(MPL)
depends=(
  gcc-libs
  readline
  sh
  zlib
)
makedepends=(
  autoconf2.13
  clang
  lld
  llvm
  python-psutil
  python-setuptools
  rust
  zip
)
checkdepends=(
  git
  mercurial
)
options=(!lto)
_relver=${pkgver}esr
source=(
  https://archive.mozilla.org/pub/firefox/releases/$_relver/source/firefox-$_relver.source.tar.xz{,.asc}
  js102-loong64-jit.patch
)
b2sums=('b70727fa91d0d270673374bebb4745b87f6194191c1c9415547d772811a4a85f79a97e8985877eb5c9beef43fe15bef574172da35935e7024a947919ec11d883'
        'SKIP'
        '2740b5d321f585cd60a038c1b1350c9e08c6559cc02a331ce7bc3d228bc785b066f795c8cbd2e43d46fe67687df420e20f4bb4bce40a2a2f962bc812aa170d01')
validpgpkeys=(
  14F26682D0916CDD81E37B6D61B7B526D98F0353  # Mozilla Software Releases <release@mozilla.com>
)

# Make sure the duplication between bin and lib is found
COMPRESSZST+=(--long)

prepare() {
  mkdir mozbuild
  cd firefox-$pkgver
  patch -p1 -i $srcdir/js102-loong64-jit.patch

  cat >../mozconfig <<END
ac_add_options --enable-application=js
mk_add_options MOZ_OBJDIR=${PWD@Q}/obj

ac_add_options --prefix=/usr
ac_add_options --enable-release
ac_add_options --enable-hardening
#ac_add_options --enable-optimize
#ac_add_options --enable-rust-simd
#ac_add_options --enable-linker=lld
ac_add_options --disable-bootstrap
ac_add_options --disable-debug
ac_add_options --disable-jemalloc
ac_add_options --disable-strip
ac_add_options --disable-unified-build

# System libraries
ac_add_options --with-system-zlib
ac_add_options --without-system-icu

# Features
ac_add_options --enable-readline
ac_add_options --enable-shared-js
ac_add_options --enable-tests
ac_add_options --with-intl-api
END
}

build() {
  cd firefox-$pkgver

  export MOZ_NOSPAM=1
  export MOZBUILD_STATE_PATH="$srcdir/mozbuild"
  export MACH_USE_SYSTEM_PYTHON=1

# clang didn't support -mlsx
  CFLAGS=${CFLAGS/-mlsx /}
  CXXFLAGS=${CFLAGS/-mlsx /}

  # Do 3-tier PGO
  echo "Building instrumented JS..."
  cat >.mozconfig ../mozconfig - <<END
#ac_add_options --enable-profile-generate=cross
END
  ./mach build

#  echo "Profiling instrumented JS..."
#  (
#    local js="$PWD/obj/dist/bin/js"
#    export LLVM_PROFILE_FILE="$PWD/js-%p-%m.profraw"
#
#    cd js/src/octane
#    "$js" run.js
#
#    cd ../../../third_party/webkit/PerformanceTests/ARES-6
#    "$js" cli.js
#
#    cd ../SunSpider/sunspider-0.9.1
#    "$js" sunspider-standalone-driver.js
#  )
#
#  llvm-profdata merge -o merged.profdata *.profraw
#
#  stat -c "Profile data found (%s bytes)" merged.profdata
#  test -s merged.profdata
#
#  echo "Removing instrumented JS..."
#  ./mach clobber
#
#  echo "Building optimized JS..."
#  cat >.mozconfig ../mozconfig - <<END
#ac_add_options --enable-lto=cross
#ac_add_options --enable-profile-use=cross
#ac_add_options --with-pgo-profile-path=${PWD@Q}/merged.profdata
#END
  ./mach build
}

check() {
  local jstests_extra_args=(
    --format=none
    --exclude-random
    --wpt=disabled
  ) jittest_extra_args=(
    --format=none
    --timeout 300
  ) jittest_test_args=(
    basic
  )

  cd firefox-$pkgver/obj
  make -C js/src check-jstests check-jit-test \
    JSTESTS_EXTRA_ARGS="${jstests_extra_args[*]}" \
    JITTEST_EXTRA_ARGS="${jittest_extra_args[*]}" \
    JITTEST_TEST_ARGS="${jittest_test_args[*]}"
}

package() {
  cd firefox-$pkgver/obj
  make DESTDIR="$pkgdir" install
  rm "$pkgdir"/usr/lib/*.ajs
  find "$pkgdir"/usr/{lib/pkgconfig,include} -type f -exec chmod -c a-x {} +
}

# vim:set sw=2 sts=-1 et:
